

# On the Conversion of Alcohol

*ABV to ABW and mole fraction and vice versa*



**A CONVERTED SPIRIT.**

*Genius of Alcohol. "AND TO THINK THAT I WAS ONCE REGARDED AS AN IMPEDIMENT TO LOCOMOTION!"*

Punch, or the London Charivari, Vol. 158, March 10th, 1920  
From the Project Gutenberg EBook of Punch

## Introduction

Accurate conversion of ABV to mole fraction or ABW is not straightforward. When ethanol and water are mixed the volumes are not additive. Mixing equal volumes of ethanol and water results in 1.92 volumes of mixture and  $100/192 = 52,1\%$  ABV instead of 50%. The purpose of this paper is to present convenient conversion formula's for ABV to ABW and mole fraction and vice versa.

## Definitions:

The Organisation Internationale de Métrologie Légale defines in Recommendation No 22: Alcoholmetry “International alcoholometric tables” alcohol strengths as follows:

### 2. Alcohol strengths.

The « alcoholic strength by volume » of a mixture of water and alcohol is the ratio of the volume of alcohol, measured at 20 °C, contained in the mixture to the total volume of the mixture, measured at the same temperature.

The « alcoholic strength by mass » of a mixture of water and alcohol is the ratio of the volume of alcohol contained in the mixture to the total mass of the mixture.

### 3. Expression of alcoholic strengths.

Alcohol strengths are expressed in parts of alcohol per hundred parts of mixture.

Their symbols are :

« % vol » for the strength by volume.

« % mass » for the strength by mass.

Alcohol strength by volume is commonly abbreviated as ABV.

Alcohol strength by mass is commonly abbreviated as ABW.

Mole fraction of an ethanol water mixture is defined as the number of molecules of ethanol divided by the total number of all molecules in the mixture.

## Symbols, units and properties

Name	Symbol	Unit
Density	$\rho$	kg/m <sup>3</sup>
Alcohol strength by mass	$p$	% mass
Alcohol strength by volume	$q$	% vol
Molar mass	$M$	g/mol
Mole fraction	$x$	

Name	Symbol	Units	Ethanol	Water	Reference
Molar mass	$M$	g/mol	46.0684	18.0153	NIST

## Alcohol strength by mass to mole fraction and vice versa

ABW is entered or retrieved as fraction, an ABW of 50% is entered or retrieved as 0.5

From ABW to mole fraction x:

$$x = \frac{p}{p + \frac{M_e}{M_w} \cdot (1 - p)}$$

or

$$x = \frac{p}{p + 2,5569 \cdot (1 - p)}$$

From mole fraction x to ABW:

$$p = \frac{x}{x + \frac{M_w}{M_e} \cdot (1 - x)}$$

or

$$p = \frac{x}{x + 0.3911 \cdot (1 - x)}$$

## Alcohol strength by volume to mass and vice versa

One of the formula's used in the recommendation No 22:



$$\rho = A_1 + \sum_{k=2}^{12} A_k p^{k-1} + \sum_{k=1}^6 B_k (t - 20^\circ\text{C})^k + \sum_{i=1}^n \sum_{k=1}^{m_i} C_{i,k} p^k (t - 20^\circ\text{C})^i$$

uses 6 vectors and 44 coefficients for calculating the density of a water alcohol mixture. This is for calculations that need no great accuracy inconvenient, so a brute force approach was used.

From Perry's chemical engineers' handbook. — 7th ed. the table 2-111 "Densities of Mixtures of C<sub>2</sub>H<sub>5</sub>OH and H<sub>2</sub>O at 20°C" was used to calculate the ABV and mole fraction for each ABW value. For the two conversions: ABW to ABV and ABV to ABW the coefficients for an octic polynomial  $y = a + b \cdot x + c \cdot x^2 + d \cdot x^3 + f \cdot x^4 + g \cdot x^5 + h \cdot x^6 + i \cdot x^7 + j \cdot x^8$  were obtained from <http://zunzun.com>. During this process a glitch in Table 2-111 was found:

ABW	Density according to OIML - R022 - Table IIIa	Density according to Perry 7 <sup>th</sup> - Table 2-111	Delta
0,715	0,86403	0,86407	0,00004
0,716	0,86379	0,86383	0,00004
0,717	0,86355	<b>0,86339</b>	<b>-0,00016</b>
0,718	0,86331	0,86335	0,00004
0,719	0,86307	0,86311	0,00004

This appears to be a typo so the density for an ABW of 0,717 was changed to 0,086359 in the input for <http://zunzun.com>. The target was fitted for the lowest peak absolute value of absolute error.

$y=a+b\cdot x+c\cdot x^2+d\cdot x^3+f\cdot x^4+g\cdot x^5+h\cdot x^6+i\cdot x^7+j\cdot x^8$		
Function	Coefficients	Absolute error
ABW to ABV	$a = -3.9705486746795932\text{E-}05$ $b = 1.2709666849144778\text{E+}00$ $c = -4.0926819348115739\text{E-}01$ $d = 2.0463351302912738\text{E+}00$ $f = -7.8964816507513707\text{E+}00$ $g = 1.5009692673927390\text{E+}01$ $h = -1.5765836469736477\text{E+}01$ $i = 8.8142267038252680\text{E+}00$ $j = -2.0695760421183493\text{E+}00$	
ABV to ABW	$a = 1.8684999875047631\text{E-}04$ $b = 7.7602465132552556\text{E-}01$ $c = 4.1803095099103116\text{E-}01$ $d = -2.5221614925275091\text{E+}00$ $f = 9.5827123045656251\text{E+}00$ $g = -1.9928886159385002\text{E+}01$ $h = 2.4165120890385651\text{E+}01$ $i = -1.5830262207383321\text{E+}01$ $j = 4.3390473620304988\text{E+}00$	

## Alcohol strength by volume to mole fraction and vice versa

Direct conversion to and from mole fraction was investigated and found to introduce a much larger error than the conversion to and from the alcohol strength by weight.

The conversion from ABV to mole fraction is therefore:  $\text{ABV} \rightarrow \text{ABW} \rightarrow \text{mole fraction}$ .

The conversion from mole fraction to ABV is therefore:  $\text{mole fraction} \rightarrow \text{ABW} \rightarrow \text{ABV}$ .

## Source code in VBA

### ***Alcohol strength by mass to mole fraction and vice versa***

```
Public Function ABW2MF(ABW)
    'conversion from alcohol by weight to mole fraction.
    'result (between 0 and 1) is only valid for an input between 0 and 1.

    Const Me = 46.0684
    Const Mw = 18.0153

    ABW2MF = ABW / (ABW + Me / Mw * (1-ABW))

End Function

Public Function MF2ABW(MF)
    'conversion from mole fraction to alcohol by weight.
    'result (between 0 and 1) is only valid for an input between 0 and 1.

    Const Me = 46.0684
    Const Mw = 18.0153

    MF2ABW = MF / (MF + Mw / Me * (1 - MF))

End Function
```

### ***Alcohol strength by volume to mass and vice versa***

```
Public Function ABW2ABV(ABW)
    'conversion from alcohol by weight to alcohol by volume.
    'result (between 0 and 1) is only valid for an input between 0 and 1.

    temp = 0.0

    ' coefficients for the octic polynomial from http://zunzun.com
    Const a = -3.9705486746795932E-05
    Const b = 1.2709666849144778E+00
    Const c = -4.0926819348115739E-01
    Const d = 2.0463351302912738E+00
    Const f = -7.8964816507513707E+00
    Const g = 1.5009692673927390E+01
    Const h = -1.5765836469736477E+01
    Const i = 8.8142267038252680E+00
    Const j = -2.0695760421183493E+00

    temp = j
    temp = temp * ABW + i
    temp = temp * ABW + h
    temp = temp * ABW + g
    temp = temp * ABW + f
    temp = temp * ABW + d
    temp = temp * ABW + c
    temp = temp * ABW + b
    temp = temp * ABW + a
    ABW2ABV = temp

End Function
```

```

Public Function ABV2ABW(ABV)
    'conversion from alcohol by volume to alcohol by weight.
    'result (between 0 and 1) is only valid for an input between 0 and 1.
    temp = 0.0

    ' coefficients for the octic polynomial from http://zunzun.com
    Const a = 1.8684999875047631E-04
    Const b = 7.7602465132552556E-01
    Const c = 4.1803095099103116E-01
    Const d = -2.5221614925275091E+00
    Const f = 9.5827123045656251E+00
    Const g = -1.9928886159385002E+01
    Const h = 2.4165120890385651E+01
    Const i = -1.5830262207383321E+01
    Const j = 4.3390473620304988E+00

    temp = j
    temp = temp * ABV + i
    temp = temp * ABV + h
    temp = temp * ABV + g
    temp = temp * ABV + f
    temp = temp * ABV + d
    temp = temp * ABV + c
    temp = temp * ABV + b
    temp = temp * ABV + a
    ABV2ABW = temp
End Function

```

## ***Alcohol strength by volume to mole fraction and vice versa***

```

Public Function ABV2MF(ABV)
    'conversion from alcohol by volume to mole fraction.
    'result (between 0 and 1) is only valid for an input between 0 and 1.

    ABV2MF = ABW2MF(ABV2ABW(ABV))
End Function

Public Function MF2ABV(MF)
    'conversion from mole fraction to alcohol by volume.
    'result (between 0 and 1) is only valid for an input between 0 and 1.

    MF2ABV = ABW2ABV(MF2ABW(MF))
End Function

```

## **Literature**

Organisation Internationale de Métrologie Légale: Recommendation No 22: Alcoholmetry “ International alcoholometric tables”

Perry’s chemical engineers’ handbook. — 7th ed. : Table 2-111 “Densities of Mixtures of C<sub>2</sub>H<sub>5</sub>OH and H<sub>2</sub>O at 20°C”

## **Revision history**

0	2013-05-12	First issue
1	2014-02-15	Corrected several errors as pointed out by Maritimer from HD forum (thank you :-))